Needham, Chapter 4, Exercise 18, p. 213.

G. Palmer, November 21, 2015

18. Recall Ex. 19, p. 186, where you showed that a general Möbius transformation

$$M(z) = \frac{az+b}{cz+d}$$

maps concentric circles to concentric circles if and only if the original circle (call if $\mathcal{F}$) is centred at q = -d/c. Let ρ = |z-q| be the distance from q to z, so that members of $\mathcal{F}$ are $\rho \equiv$ const.

```
In[8268]:= Clear["Global`*"]

(* PROGRAM FOR FIGURE 1. CONCENTRIC CIRCLES UNDER M(Z). Added November 15,
2015, revised November 21, 2015. *)

(* CONSTANTS. Keep a,b,c,d within narrow range
 to avoid imploding or exploding. *)

(* In this version, ρ = |w|, and z = w + q, where q = -d/c. *)

pr = 2;

(* a, b, c, d, and absϵ are all arbitrarily chosen, but it makes
 the plot more manageable if a, b, c, and d have similar absolute values. *)

a = Exp[I Pi / 3];

b = Exp[I 1.9 Pi / 3];

c = Exp[I 4 Pi / 3];

d = Exp[I 5 Pi / 4];

q = -d / c;

absϵ = .3 ; (* All the ϵ vectors in ℱ
 are same size and direction (blue arrows). *)

cNumbs = {0, 1, a, b, c, d};

nCircles = 15;

firstCirc = 3; (* first circle displayed *)

lastCirc = 8; (* last circle displayed *)


(* FUNCTIONS *)

v[z_] := {Re @ z, Im @ z}; (* Complex to vector *)

m[z_] := (a z + b) / (c z + d); (* the Möbius transformation *)
```

```
SetAttributes[{v, m}, Listable]; (* enables v and m to process lists *)

(* amplification = |M'(w)| |w| = ρ *)
(* For this problem, twist is applied to points in circles centered at origin.
  For convience, the points are usually calculated as nϵ, n a positive integer,
ϵ a small vector. w = z-q = nϵ, n a pos. integer indexing a circle in ℱ,
ϵ a small vector orthogonal to a circle in ℱ. *)

ampli[w_] := Abs[(a d - b c)] / (Abs[c]^2 Abs[w]^2);

(* twist = Arg[M'(z)]; z is point on circle centered at q *)
(* w = z-q is an origin-centered point from which ϵ emanates. *)

twist[w_] := Arg[(a d - b c) / ((c )^2 w^2)];

(* w is point from which ϵ emanates; w = z-q *)

amplitwist[w_] := ampli[w] Exp[I twist[w]];


(* GRAPHICS OBJECTS *)

    (* POINTS *)

points = Point @ v @ cNumbs;

qPt = {PointSize @ .01, Point @ v @ q};

acPt = {Gray, PointSize @ .01, Point @ v[a / c]};

points = {points, acPt, qPt};


    (* LINES *)

xAxis = {Gray, Line @ v @ {-2 pr, 2 pr}};

yAxis = {Gray, Line @ v @ {-pr I, 2 pr I}};

size = .01;

θArrows = 3 Pi / 4;  (* arbitrary angle for arrows on ℱ *)

ϵ = absϵ Exp[I θArrows]; (* a small complex number for a vector *)

eArrows = Table[ {Blue, Arrowheads @ size, Arrow[v @ {n ϵ + q, (n + 1) ϵ + q}]},
  {n, Range[1, nCircles]}];

(* amplitwisted ϵ arrows at w + q, where w = nϵ *)
atϵArrows = Table[ {Green, Arrowheads @ size,

     Arrow[v @{

        m [n ϵ + q], (* point of emanation of vector under M(z) *)
```

```
        m [n ϵ + q] + amplitwist[n ϵ] ϵ  (* w = nϵ *)

       }]},

   {n, Range[1, nCircles - 1]}];


lines = {xAxis, yAxis,
   eArrows[[firstCirc ;; lastCirc]], atϵArrows[[firstCirc ;; lastCirc]]};


    (* ARCS *)

(* points for circles in ℱ *)
theℱPts =
  Table[Table[(n Abs @ ϵ) Exp[I θ] + q, {θ, 0, 2.1 Pi, .01}], {n, Range[1, nCircles]}];

(* points for circles in M(ℱ) *)
mℱPts = Table[Table[m @ pt, {pt, circle}], {circle, theℱPts}];

(* ℱ *)
zCircles = Table[{Black, Line[v @ circPts]}, {circPts, theℱPts}];

(* M(ℱ) *)
mCircles = Table[{LightGray, Thick, Line[v @ circle]}, {circle, mℱPts}];

arcs = {zCircles[[firstCirc ;; lastCirc]], mCircles[[firstCirc ;; lastCirc]]};


    (* POLYGONS *)

scale = .2;

nTriangles = 8;  (* number of triangles *)

cosmeticRotation = Pi / 1.5; (* applied to ϵ shapes *)

vertices = scale {0, .8 I, Sqrt @ 2}; (* complex vertex points of one triangle *)

(* complex points of n triangles at increasing distance,
ρ = n|ϵ|, where w = nϵ *)
triPtsSet = Table[ vertices + n ϵ + q, {n, Range[1, nTriangles]}];

(* graphics objects for triangles *)
triangles = Table[ {FaceForm @ LightBlue, EdgeForm @ Black, Polygon[v @ vertices]},
   {vertices, triPtsSet}];

(* rotate triangles out of way of arrows *)
 triangles = Rotate[triangles, cosmeticRotation, v @ q];


(* points of triangles under M at w+q with increasing ρ = n|ϵ| in ℱ *)
mTriPtsSet = Table[Table[m[n ϵ + q] + amplitwist[ n ϵ ] vertex, {vertex, vertices}] ,
   {n, Range[1, nTriangles]}];
```

```
mTriangles = Table[ {FaceForm @ LightGreen, EdgeForm @ Black, Polygon[v @ mTriPts]},
   {mTriPts, mTriPtsSet}];

mTriangles = Rotate[mTriangles, cosmeticRotation, v[a / c]];


polygons = {triangles, mTriangles};


(* TEXT *)

adj = .1 + .1 I;
default = 12;
small = 8;
preTwist = (a d - b c) / c^2;
preAmpli = Abs[preTwist];

txtData = {

   {"0", -adj, 1.5 default},
   {"1", 1 + adj, default},
   {"a", a + adj, default},
   {"b", b + adj, default},
   {"c", c + adj, default},
   {"d", d + adj, default},
   {"q", q + adj, default},
   {"a/c", a / c - .15 I, default},
   {"Figure 1. ℱ (black) under M(z) = (az+b)/(ac+d) (gray)" <>
     "\namplification = " <> ToString[Abs[preAmpli / Abs[c]^2]] <>
     "/ρ² = |M'(z)|" <> "\ntwist = (" <> ToString[Re[preTwist]] <> "+" <>
     ToString[Im[preTwist]] <> "i)/w² = arg(M'(z))", Re[a / c] - 1 - 2.5 pr I, default}
  };

(* labels for numbers on preimages *)

labels = Table[ Text[Style[item[[1]], item[[3]]], v @ item[[2]]], {item, txtData}];

numbers =
  Table[ {ToString[n], (nTriangles - (n - 1)) ϵ + q}, {n, Range[1, nTriangles]}];

numberLabels =
  Table[ Text[Style[item[[1]], small, Bold], v @ item[[2]]], {item, numbers}];

numberLabels =
  Table[Rotate[text, cosmeticRotation - Pi / 20, v @ q], {text, numberLabels}];

numberLabels = Table[Rotate[text, - cosmeticRotation], {text, numberLabels}];


(* labels for numbers on images M[shape] *)

mNumbers =
  Table[ {ToString[n], m[(nTriangles - (n - 1)) ϵ + q]}, {n, Range[1, nTriangles]}];
```

```
mNLabels =
  Table[ Text[Style[item[[1]], small, Bold], v @ item[[2]]], {item, mNumbers}];

mNLabels = Table[Rotate[text, 1.45 (Pi / 2), v[a / c]], {text, mNLabels}];

mNLabels = Table[Rotate[text, -1.45 (Pi / 2)], {text, mNLabels}];

labels = labels ~ Join ~ numberLabels ~ Join ~ mNLabels;


(* GRAPHICS *)

g1 = Graphics[{lines, arcs, polygons, labels, points}, Background → White,
  PlotRange → {{-2 pr, 1.1 pr}, {- 3.1 pr, 1.5 pr}}, ImageSize → Large]
```

Out[8320]=



Figure 1. $\mathcal{F}$ (black) under M(z) = (az+b)/(ac+d) (gray)
amplification = 1.13281/ρ^2 = |M'(z)|
twist = (−0.378141+1.06784i)/w^2 = arg(M'(z))

(i) By considering orthogonal connecting vectors from one member of $\mathcal{F}$ to an infinitesimally larger member of $\mathcal{F}$, deduce that the amplification of M is constant on each circle of $\mathcal{F}$. Deduce that |M'| must be a function of ρ alone.

The circles of $\mathcal{F}$ are concentric, so orthogonal ϵ arrows connecting $\mathcal{F}_n$ to $\mathcal{F}_{n-1}$ have the same length on the entire circumference of $\mathcal{F}_n$. M preserves angles, so an orthogonal ϵ arrow on $\mathcal{F}$ goes to an orthogonal ϵ arrow on $\tilde{\mathcal{F}}$.

Let w = M(z) = $\tilde{\rho}\, e^{i\theta}$. We can expect $\tilde{\rho}$ to be constant across all θ because the circles of $\tilde{\mathcal{F}}$ are concentric. We also expect $\tilde{\epsilon}$ to be constant across all θ for the same reason. If $\tilde{\rho}_\theta$ were different from $\tilde{\rho}_{\theta+\phi}$ on a circle of $\tilde{\mathcal{F}}$, then orthogonal angles would not be preserved. Hence, the amplification of M is constant on each circle of $\mathcal{F}$ and |M'| must be a function of ρ alone, as opposed to ρ and θ.

We attempt to deduce this by taking the derivative of M(z).

$$M'(z) = \left(\frac{az+b}{cz+d}\right)' = \left((az+b)\,(cz+d)^{-1}\right)'$$

$$= a(cz+d)^{-1} + (az+b)\,(-c)\,(cz+d)^{-2}$$

$$= \frac{a(cz+d)-c(az+b)}{(cz+d)^2} = \frac{acz+ad-(acz+bc)}{c^2(z-q)^2} = \frac{ad-bc}{c^2(z-q)^2} \qquad (1)$$

We use (1) in (vi) and in programming twist in Figure 1.

$$|M'(z)| = \frac{|ad-bc|}{\left|c^2(z-q)^2\right|} = \frac{|ad-bc|}{|c|^2\rho^2}$$

Points a, b, c and d are constants. Hence, for M such that (ad-bc) ≠ 0 and c ≠ 0, |M'| depends on ρ alone.

(ii) By considering the image of an infinitesimal shape that starts far from q and then travels to a point very close to q, deduce that at some point in the journey, the image and preimage are congruent.

[November 17, 2015. I think there is a problem with the question. Given the distinction between image and preimage, if the image is M($\mathcal{F}$), then it does not naturally travel to a point very close to q by decreasing ρ, which is a quality of the preimage, but the image may approach a/c with increasing ρ. Therefore, I think the question should read *"By considering the image under M of an infinitesimal shape that starts far from q and then travels to a point very close to q, deduce that at some point in the journey, the image and preimage are congruent."]*

Stipulate that the image is M(P), where P is the preimage, z is a point in the preimage, and ρ = |z-q| = |z+d/c|. Let ϵ be the vector of any side of the preimage or a tangent emanating from a point on the preimage and $\tilde{\epsilon}$ be the corresponding side or tangent on the image. M is conformal, because it is a Möbius transformation. Angles are preserved. It has amplitwist because it is conformal over a region. Since it has amplitwist, the image at any size is similar to the preimage. We have shown in (i) that the amplification |M'| depends only on ρ. Furthermore, Vasco has called attention to the fact that as ρ decreases, $\tilde{\epsilon}$ vectors increase (See (i)). In this case, we are interested in decreasing ρ. When ρ is very large, $|\tilde{\epsilon}|$ is very near zero. When ρ is very small, $|\tilde{\epsilon}|$ is large. During the journey, which is measured by decreasing ρ, $|\tilde{\epsilon}|$ increases until $|\tilde{\epsilon}| = |\epsilon|$, at which point |M'(z)| = 1 for all z, and the images are congruent.

Discussion of Figure 1: Figure 1 illustrates these points. The triangles represent ϵ shapes. The outermost preimage triangle (blue) corresponds to the innermost image triangle (green). Because M is conformal, these images are similar. As ρ decreases for the preimage triangles, $\tilde{\rho} = |\tilde{z}$-(a/c)| increases for the image triangles and they increase in size. The triangle that would be congruent with the preimage triangles would appear somewhere between the fifth and sixth $\tilde{\mathcal{F}}$ image circle from the center. This is where |M'| ≈ 1. We have calculated the twist as arg(M'(z)) = arg((ad − bc)$/(c^2 w^2)$), where w = z-q, without normalization. Both the preimage and image triangles were rotated by Pi/2 around q and a/c respectively after the calculations just to get them out of the way of the ϵ arrows, but by the question, they can be situated at any arg(z) and arg(M(z)).

Vasco also pointed out that the twist can be rewritten as $\arg(A/w^2)$, where A = (ad − bc)$/c^2$) = $ge^{i\alpha}$, so $\arg(A/w^2)$ = -2ϕ + α and ϕ = arg(w). We can also see by Figure[1], p. 124, and by obtaining step (iii) in (3) on the same page, that M(z) has the rotation arg(-A/w) = -(α -2ϕ) = -2ϕ + α + Pi. (Step 4 would not alter the rotation or twist.) This shows that the twist of M reverses the direction of the arrows.

(iii) Combine the above results to deduce that there is a special member I_M of $\mathcal{F}$ such that infinitesimal shapes on I_M are mapped to congruent image shapes on the image circle $M(I_M)$. Recall that I_M is called the isometric circle of M.

I_M is the member at which |M'(z)| = 1 for all z in I_M.

(iv) Use the previous part to explain why $M(I_M)$ has the same radius as I_m.

The amplification depends only on ρ. Since $|\tilde{\epsilon}| = |\epsilon|$, it must be the case that M'(z) = M'(w) = 1, which implies that (w = M(z)) = z, and since ρ = |z-q| and $\tilde{\rho}$ = |w-q|, it follows that $\tilde{\rho} = \rho$.

(v) Explain why $I_{M^{-1}} = M(I_M)$.

$I_M = M(I_M)$, from (iv), because they have the same radius.

$I_{M^{-1}} = M^{-1}(I_M) = M^{-1} \circ M(I_M) = I_M = M(I_M)$

(vi) Suppose that M is normalized. Using the idea in Ex. 16, show that the amplification of M is

$$|M'(z)| = \frac{1}{|c|^2 \rho^2}$$

From Ex. 16, the amplification a = |f'(z)|.

$$|M'(z)| = \frac{|ad - bc|}{|c|^2 \rho^2} \qquad \text{from (i)}$$

If M is normalized, ad-bc = 1. Then, letting k = ± $\sqrt{ad - bc}$, there is a $\left|\frac{c}{k}\right|^2$ in the denominator, so the k goes to 1.

$$|M'(z)| = \frac{1}{|c|^2 \rho^2}$$